In[1]:= 
```
NotebookDirectory[]
```

Out[1]= 
```
C:\Dropbox\307_Files\2019\
```

Before reading this notebook evaluate the entire notebook: the shortcut Alt+v+o or the menu item:

Evaluation ▹ Evaluate Notebook

## A symmetric tree

A picture of a “symmetric” Pythagorean theorem is completely determined by two end-points of the hypotenuse. The following command calculates ending points of two smaller squares given the end-points of the hypotenuse.

In[2]:= 
```
Reverse[{x, y}] {-1, 1}
```

Out[2]= 
```
{-y, x}
```

In[3]:= 
```
Clear[Pythagoras, pA, pB];
Pythagoras[{pA_, pB_}] := Module[
  {pC, pD, pE, pF, pG},
  pC = 1/2 (pA + pB) + Reverse[1/2 (pB - pA)] {-1, 1};
  pD = pC + Reverse[(pC - pA)] {-1, 1}; pE = pD - (pC - pA);
  pF = pB + Reverse[(pB - pC)] {-1, 1}; pG = pF + (pC - pB); {{pE, pD}, {pG, pF}}
 ]
```

Testing

In[5]:= 
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
```

In[6]:= 
```
Pythagoras[pts]
```

Out[6]= $\left\{\left\{\left\{-1, \frac{1}{2}\right\}, \left\{-\frac{1}{2}, 1\right\}\right\}, \left\{\left\{\frac{1}{2}, 1\right\}, \left\{1, \frac{1}{2}\right\}\right\}\right\}$

In[7]:= 
```
Point[Flatten[Pythagoras[pts], 1][[#]]] & /@ Range[4](*mky own test to see what it does*)
```

Out[7]= $\left\{\text{Point}\left[\left\{-1, \frac{1}{2}\right\}\right], \text{Point}\left[\left\{-\frac{1}{2}, 1\right\}\right], \text{Point}\left[\left\{\frac{1}{2}, 1\right\}\right], \text{Point}\left[\left\{1, \frac{1}{2}\right\}\right]\right\}$

In[8]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};

Graphics[
 {
  {Green, Thickness[0.01], Line[pts]},
  {{Text[#, Flatten[Pythagoras[pts], 1]⟦#⟧, {1, 1}],
      PointSize[0.01], Point[Flatten[Pythagoras[pts], 1]⟦#⟧]} & /@ Range[4]}
 },
 {AspectRatio → Automatic, PlotRange → {{-3, 3}, {-1.5, 3}}}, Frame → True
]
```

Out[9]=




A miracle of these points is that we can easily reconstruct the picture of the theorem from them using the square making command below

In[10]:=
```
Clear[MySquare, pA, pB];

MySquare[{pA_, pB_}] := Module[{pC, pD},
  pC = pB - Reverse[(pB - pA)] {-1, 1}; pD = pC + pA - pB; Polygon[{pA, pB, pC, pD, pA}]
 ]
```

In[12]:=
```
MySquare[{{0, 0}, {1, 0}}]
```

Out[12]=
```
Polygon[{{0, 0}, {1, 0}, {1, -1}, {0, -1}, {0, 0}}]
```

In[13]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
Graphics[
 {
  {Green, Thickness[0.01], Line[pts]}, {Green, MySquare[pts]},
  {{Text[#, Flatten[Pythagoras[pts], 1][[#]], {1, 1}],
      PointSize[0.01], Point[Flatten[Pythagoras[pts], 1][[#]]]} & /@
    Range[4]}, {Red, Map[MySquare, Pythagoras[pts]]}
 },
 {AspectRatio → Automatic, PlotRange → {{-3, 3}, {-1.5, 3}}, Frame → True
]
```

Out[14]=




The only remaining thing is to nest this command to produce more squares.

In[15]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
```

We start with a pair of points and produce two pairs of points which are in a list.

In[16]:=
```
Pythagoras[pts]
```

Out[16]= $\left\{\left\{\left\{-1, \frac{1}{2}\right\}, \left\{-\frac{1}{2}, 1\right\}\right\}, \left\{\left\{\frac{1}{2}, 1\right\}, \left\{1, \frac{1}{2}\right\}\right\}\right\}$

It is important to notice that the input and the output are of different structure. The input is just a pair of points, while the output is a list of pairs. However, we can map our function Pythagoras[] to this list of pairs.

In[17]:=
```
Map[Pythagoras[#] &, Pythagoras[pts]]
```

Out[17]= $\left\{\left\{\left\{\left\{-\frac{3}{2}, \frac{1}{2}\right\}, \left\{-\frac{3}{2}, 1\right\}\right\}, \left\{\left\{-1, \frac{3}{2}\right\}, \left\{-\frac{1}{2}, \frac{3}{2}\right\}\right\}\right\}, \left\{\left\{\left\{\frac{1}{2}, \frac{3}{2}\right\}, \left\{1, \frac{3}{2}\right\}\right\}, \left\{\left\{\frac{3}{2}, 1\right\}, \left\{\frac{3}{2}, \frac{1}{2}\right\}\right\}\right\}\right\}$

Unfortunately what we get here is even worse structure. It is a list of lists of pairs of points. Ideally we would want to get just a list of pairs of points.

In[18]:=
```
Map[Pythagoras[#] &, Pythagoras[pts]][[1]]
```

Out[18]= $\left\{\left\{\left\{-\frac{3}{2}, \frac{1}{2}\right\}, \left\{-\frac{3}{2}, 1\right\}\right\}, \left\{\left\{-1, \frac{3}{2}\right\}, \left\{-\frac{1}{2}, \frac{3}{2}\right\}\right\}\right\}$

To get a list of pairs we Flatten[]

In[19]:=
```
Flatten[Map[Pythagoras[#] &, Pythagoras[pts]], 1][[1]]
```

Out[19]= $\left\{\left\{-\frac{3}{2}, \frac{1}{2}\right\}, \left\{-\frac{3}{2}, 1\right\}\right\}$

In[20]:=
```
Flatten[Map[Pythagoras[#] &, Pythagoras[pts]], 1]
```

Out[20]= $\left\{\left\{\left\{-\frac{3}{2}, \frac{1}{2}\right\}, \left\{-\frac{3}{2}, 1\right\}\right\}, \left\{\left\{-1, \frac{3}{2}\right\}, \left\{-\frac{1}{2}, \frac{3}{2}\right\}\right\}, \left\{\left\{\frac{1}{2}, \frac{3}{2}\right\}, \left\{1, \frac{3}{2}\right\}\right\}, \left\{\left\{\frac{3}{2}, 1\right\}, \left\{\frac{3}{2}, \frac{1}{2}\right\}\right\}\right\}$

Now we organize this in a function which will take a **list of pairs of points** as an input and output a bigger **list of pairs of points**.

In[21]:=
```
(* lis is a list of pairs of points *)

Clear[LPythagoras, lis];

LPythagoras[lis_] := Flatten[Map[Pythagoras[#] &, lis], 1]
```

In[23]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
```

Above we defined pts to be a pair of points. Since we need a **list** of pairs of points we need to wrap pts in braces.

In[24]:=
```
LPythagoras[{pts}]
```

Out[24]= $\left\{\left\{\left\{-1, \frac{1}{2}\right\}, \left\{-\frac{1}{2}, 1\right\}\right\}, \left\{\left\{\frac{1}{2}, 1\right\}, \left\{1, \frac{1}{2}\right\}\right\}\right\}$

In this way we have obtained a function which can be composed with itself.

In[25]:=
```
LPythagoras[LPythagoras[{pts}]]
```

Out[25]= $\left\{\left\{\left\{-\frac{3}{2}, \frac{1}{2}\right\}, \left\{-\frac{3}{2}, 1\right\}\right\}, \left\{\left\{-1, \frac{3}{2}\right\}, \left\{-\frac{1}{2}, \frac{3}{2}\right\}\right\}, \left\{\left\{\frac{1}{2}, \frac{3}{2}\right\}, \left\{1, \frac{3}{2}\right\}\right\}, \left\{\left\{\frac{3}{2}, 1\right\}, \left\{\frac{3}{2}, \frac{1}{2}\right\}\right\}\right\}$

In[26]:=
```
LPythagoras[LPythagoras[LPythagoras[{pts}]]]
```

Out[26]= $\left\{\left\{\left\{-\frac{7}{4}, \frac{1}{4}\right\}, \left\{-2, \frac{1}{2}\right\}\right\}, \left\{\{-2, 1\}, \left\{-\frac{7}{4}, \frac{5}{4}\right\}\right\}, \left\{\left\{-\frac{5}{4}, \frac{7}{4}\right\}, \{-1, 2\}\right\}, \left\{\left\{-\frac{1}{2}, 2\right\}, \left\{-\frac{1}{4}, \frac{7}{4}\right\}\right\},$
$\left\{\left\{\frac{1}{4}, \frac{7}{4}\right\}, \left\{\frac{1}{2}, 2\right\}\right\}, \left\{\{1, 2\}, \left\{\frac{5}{4}, \frac{7}{4}\right\}\right\}, \left\{\left\{\frac{7}{4}, \frac{5}{4}\right\}, \{2, 1\}\right\}, \left\{\left\{2, \frac{1}{2}\right\}, \left\{\frac{7}{4}, \frac{1}{4}\right\}\right\}\right\}$

In[27]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};

Flatten[Nest[LPythagoras[#] &, {pts}, 5], 1]
```

Out[28]=

$$\left\{\left\{-\frac{13}{8}, -\frac{1}{8}\right\}, \left\{-\frac{7}{4}, -\frac{1}{4}\right\}, \left\{-2, -\frac{1}{4}\right\}, \left\{-\frac{17}{8}, -\frac{1}{8}\right\}, \left\{-\frac{19}{8}, \frac{1}{8}\right\}, \left\{-\frac{5}{2}, \frac{1}{4}\right\}, \left\{-\frac{5}{2}, \frac{1}{2}\right\},\right.$$
$$\left\{-\frac{19}{8}, \frac{5}{8}\right\}, \left\{-\frac{19}{8}, \frac{7}{8}\right\}, \left\{-\frac{5}{2}, 1\right\}, \left\{-\frac{5}{2}, \frac{5}{4}\right\}, \left\{-\frac{19}{8}, \frac{11}{8}\right\}, \left\{-\frac{17}{8}, \frac{13}{8}\right\}, \left\{-2, \frac{7}{4}\right\}, \left\{-\frac{7}{4}, \frac{7}{4}\right\},$$
$$\left\{-\frac{13}{8}, \frac{13}{8}\right\}, \left\{-\frac{13}{8}, \frac{13}{8}\right\}, \left\{-\frac{7}{4}, \frac{7}{4}\right\}, \left\{-\frac{7}{4}, 2\right\}, \left\{-\frac{13}{8}, \frac{17}{8}\right\}, \left\{-\frac{11}{8}, \frac{19}{8}\right\}, \left\{-\frac{5}{4}, \frac{5}{2}\right\},$$
$$\left\{-1, \frac{5}{2}\right\}, \left\{-\frac{7}{8}, \frac{19}{8}\right\}, \left\{-\frac{5}{8}, \frac{19}{8}\right\}, \left\{-\frac{1}{2}, \frac{5}{2}\right\}, \left\{-\frac{1}{4}, \frac{5}{2}\right\}, \left\{-\frac{1}{8}, \frac{19}{8}\right\}, \left\{\frac{1}{8}, \frac{17}{8}\right\}, \left\{\frac{1}{4}, 2\right\},$$
$$\left\{\frac{1}{4}, \frac{7}{4}\right\}, \left\{\frac{1}{8}, \frac{13}{8}\right\}, \left\{-\frac{1}{8}, \frac{13}{8}\right\}, \left\{-\frac{1}{4}, \frac{7}{4}\right\}, \left\{-\frac{1}{4}, 2\right\}, \left\{-\frac{1}{8}, \frac{17}{8}\right\}, \left\{\frac{1}{8}, \frac{19}{8}\right\}, \left\{\frac{1}{4}, \frac{5}{2}\right\},$$
$$\left\{\frac{1}{2}, \frac{5}{2}\right\}, \left\{\frac{5}{8}, \frac{19}{8}\right\}, \left\{\frac{7}{8}, \frac{19}{8}\right\}, \left\{1, \frac{5}{2}\right\}, \left\{\frac{5}{4}, \frac{5}{2}\right\}, \left\{\frac{11}{8}, \frac{19}{8}\right\}, \left\{\frac{13}{8}, \frac{17}{8}\right\}, \left\{\frac{7}{4}, 2\right\}, \left\{\frac{7}{4}, \frac{7}{4}\right\},$$
$$\left\{\frac{13}{8}, \frac{13}{8}\right\}, \left\{\frac{13}{8}, \frac{13}{8}\right\}, \left\{\frac{7}{4}, \frac{7}{4}\right\}, \left\{2, \frac{7}{4}\right\}, \left\{\frac{17}{8}, \frac{13}{8}\right\}, \left\{\frac{19}{8}, \frac{11}{8}\right\}, \left\{\frac{5}{2}, \frac{5}{4}\right\}, \left\{\frac{5}{2}, 1\right\}, \left\{\frac{19}{8}, \frac{7}{8}\right\},$$
$$\left.\left\{\frac{19}{8}, \frac{5}{8}\right\}, \left\{\frac{5}{2}, \frac{1}{2}\right\}, \left\{\frac{5}{2}, \frac{1}{4}\right\}, \left\{\frac{19}{8}, \frac{1}{8}\right\}, \left\{\frac{17}{8}, -\frac{1}{8}\right\}, \left\{2, -\frac{1}{4}\right\}, \left\{\frac{7}{4}, -\frac{1}{4}\right\}, \left\{\frac{13}{8}, -\frac{1}{8}\right\}\right\}$$

In[29]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
nn = 5;
Graphics[
 {
  {Green, Thickness[0.01], Line[pts]},
  {{Text[#, Flatten[Nest[LPythagoras[#] &, {pts}, nn], 1][[#]], {1, 1}], PointSize[0.01],
      Point[Flatten[Nest[LPythagoras[#] &, {pts}, nn], 1][[#]]]} & /@ Range[2^(nn + 1)]}
 },
 {AspectRatio → Automatic, PlotRange → {{-4, 4}, {-2, 4}}}, Frame → True
]
```

Out[31]=



In[32]:=
```
Sin[#] & /@ Range[1, 10]
```

Out[32]=
```
{Sin[1], Sin[2], Sin[3], Sin[4], Sin[5], Sin[6], Sin[7], Sin[8], Sin[9], Sin[10]}
```

In[33]:=
```
Map[Sin[#] &, Range[1, 10]]
```

Out[33]=
```
{Sin[1], Sin[2], Sin[3], Sin[4], Sin[5], Sin[6], Sin[7], Sin[8], Sin[9], Sin[10]}
```

In[34]:=
```
ColorData["Indexed"][[2]]
```

Out[34]=
```
2
```

In[35]:=
```
ColorData[3, "ColorList"]⟦2⟧
```

Out[35]=
```
RGBColor[0.996078, 0.360784, 0.027451]
```

In[36]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
Clear[nn];
Graphics[
 {
  {Green, MySquare[pts]},
  {Table[{ColorData[3, "ColorList"]⟦nn⟧,
     Map[MySquare[#] &, Nest[LPythagoras[#] &, {pts}, nn]]}, {nn, 1, 4}]}
 },
 {AspectRatio → Automatic, PlotRange → {{-4, 4}, {-2, 4}}}, Frame → True
]
```

Out[38]=



In[74]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};
Clear[nn];
Manipulate[Graphics[
  {
   {Green, MySquare[pts]},
   {Table[{ColorData[3, "ColorList"][[nn]],
      Map[MySquare, Nest[LPythagoras[#] &, {pts}, nn]]}, {nn, 1, mm}]}
  },
  {AspectRatio → Automatic, PlotRange → {{-4, 4}, {-2, 4}}}, Frame → True
 ], {{mm, 6}, 1, 10, 1, ControlType → Setter, ControlPlacement → Top}]
```

Out[76]=




# A general tree

First figure out rotation by $\theta$. The matrix is

In[42]:=
```
{{Cos[th], -Sin[th]}, {Sin[th], Cos[th]}}
```

Out[42]=
```
{{Cos[th], -Sin[th]}, {Sin[th], Cos[th]}}
```

I just incorporated angle in all commands above.

In[43]:=
```
Clear[PythagorasA, pA, pB, th];

PythagorasA[{pA_, pB_}, th_] := Module[{pC, pD, pE, pF, pG},
  pC = N[1/2 (pA + pB) + {{Cos[th], -Sin[th]}, {Sin[th], Cos[th]}}.(1/2 (pB - pA))];
  pD = pC + Reverse[(pC - pA)] {-1, 1}; pE = pD - (pC - pA);
  pF = pB + Reverse[(pB - pC)] {-1, 1}; pG = pF + (pC - pB); N[{{pE, pD}, {pG, pF}}]
 ]
```

In[45]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; th = Pi / 3;
Graphics[
 {
  {Green, Thickness[0.01], Line[pts]},
  {{Text[#, Flatten[PythagorasA[pts, th], 1][[#]], {1, 1}],
      PointSize[0.01], Point[Flatten[PythagorasA[pts, th], 1][[#]]]} & /@ Range[4]}
 },
 {AspectRatio → Automatic, PlotRange → {{-3, 3}, {0, 3}}}, Frame → True
]
```

Out[46]=




In[47]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};

th = Pi / 3;

Map[PythagorasA[#, th] &, Pythagoras[pts]][[1]]
```

Out[49]=
```
{{{-1.59151, 0.658494}, {-1.43301, 1.25}}, {{-0.75, 1.43301}, {-0.408494, 1.34151}}}
```

In[50]:=
```
(* lis is a list of pairs of points *)
Clear[LPythagorasA, lis, th];

LPythagorasA[lis_, th_] := Flatten[Map[PythagorasA[#, th] &, lis], 1]
```

In[52]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; th = Pi / 3;

Flatten[Nest[LPythagorasA[#, th] &, {pts}, 3], 1]
```

Out[53]=
```
{{-2.23205, 1.125}, {-2.23205, 1.77452}, {-1.58253, 2.14952}, {-1.20753, 2.14952},
 {-0.341506, 2.14952}, {0.0334936, 2.14952}, {0.25, 1.77452}, {0.25, 1.55801},
 {0.899519, 1.43301}, {1.27452, 1.43301}, {1.49103, 1.05801}, {1.49103, 0.841506},
 {1.49103, 0.341506}, {1.49103, 0.125}, {1.27452, 1.11022×10^-16}, {1.14952, 1.11022×10^-16}}
```

In[54]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; th = Pi / 3;
nn = 2;
Graphics[
 {
  {Green, Thickness[0.01], Line[pts]},
  {{Text[#, Flatten[Nest[LPythagorasA[#, th] &, {pts}, nn], 1][[#]], {1, 1}], PointSize[0.01],
      Point[Flatten[Nest[LPythagorasA[#, th] &, {pts}, nn], 1][[#]]]} & /@ Range[2^(nn + 1)]}
 },
 {AspectRatio → Automatic, PlotRange → {{-4, 4}, {-2, 4}}}, Frame → True
]
```

Out[56]=




In[57]:=
```
Clear[MySquare, pA, pB];

MySquare[{pA_, pB_}] := Module[{pC, pD},
  pC = pB - Reverse[(pB - pA)] {-1, 1}; pD = pC + pA - pB; Polygon[{pA, pB, pC, pD, pA}]
 ]
```

In[59]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; th = Pi / 3;
Clear[nn];
Graphics[
 {
  {Green, MySquare[pts]},
  {Table[{ColorData[3, "ColorList"]〚nn〛,
     Map[MySquare, Nest[LPythagorasA[#, th] &, {pts}, nn]]}, {nn, 1, 5}]}
 },
 {AspectRatio → Automatic, PlotRange → {{-4.5, 4.5}, {-1.5, 3.5}}}, Frame → True
]
```

Out[61]=



In[77]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}};

Clear[nn]; angle = Pi / 3;

Manipulate[Graphics[
  {
   {Green, MySquare[pts]},
   {Table[{ColorData[3, "ColorList"][[nn]],
      Map[MySquare, Nest[LPythagorasA[#, angle] &, {pts}, nn]]}, {nn, 1, mm}]}
  },
  {AspectRatio → Automatic, PlotRange → {{-4.5, 4.5}, {-1.5, 3.5}}, Frame → True
 ], {{mm, 3}, 1, 10, 1, ControlType → Setter, ControlPlacement → Top}]
```

Out[79]=



In[80]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; th = Pi / 3; Clear[th];
Clear[nn];
Manipulate[Graphics[
  {
   {Green, MySquare[pts]},
   {Table[{ColorData[3, "ColorList"][[nn]],
      Map[MySquare, Nest[LPythagorasA[#, th] &, {pts}, nn]]}, {nn, 1, 7}]}
  },
  {AspectRatio → Automatic, PlotRange → {{-4.5, 4.5}, {-1.5, 3.5}}}, Frame → True
 ], {{th, Pi / 3}, 0, Pi, ControlType → VerticalSlider, ControlPlacement → Right}]
```

Out[82]=



In[83]:=
```
pts = {{-1 / 2, 0}, {1 / 2, 0}}; Clear[th];
Clear[nn];

Manipulate[Graphics[
  {
   {Green, MySquare[pts]},
   {Table[{ColorData[3, "ColorList"]⟦nn⟧,
      Map[MySquare, Nest[LPythagorasA[#, th] &, {pts}, nn]]}, {nn, 1, mm}]}
  },
  {AspectRatio → Automatic, PlotRange → {{-4.5, 4.5}, {-1.5, 3.5}}, Frame → True
 ], {{mm, 6}, Range[10], ControlType → Setter},
 {{th, Pi / 3}, 0, Pi, ControlType → VerticalSlider, ControlPlacement → Right}]
```

Out[85]=